Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 18

Double Integrals over General Regions

Initializations

```
> restart;
  with(plots):
  with(student):
>
```

18.1 Double Integrals over General Regions

When we evaluate double integrals over non-rectangular regions, then the limits on the inner integral are no longer constant, they become functions of the integration variable in the outer integral.

Examples

Example 18.1.1

Integrate $f(x, y) = 2x^2 y + 5x - 7y^3$ over the triangular region bounded by the coordinate axes and the line x + 3y = 6.

Solution
First we make a sketch of the region of integration. For purpose of referral, it is advised to assign variable names to the different parts of the boundary.

```
> f:=(x, y)->2*x^2*y+5*x-7*y^3;
  bnd1:=x=0;
  bnd2:=y=0;
  bnd3:=x+3*y=6;
```

$$f := (x, y) \rightarrow 2x^2 y + 5x - 7y^3$$

$$bnd1 := x = 0$$

$$bnd2 := y = 0$$

$$bnd3 := x + 3y = 6 \qquad \textbf{(2.1.1.1)}$$

```
> implicitplot({seq(bnd||k, k=1..3)}, x=-1..8, y=-1..3,
  axes=boxed);
```



Like in the previous section we have two options:

a) First integrate over *y*, then over *x*.
b) First integrate over *x*, then over *y*.

Which approach is easier depends both on the integrant and on the integration region. In this case there is no clear advantage to choose one method above the other. We will first integrate over y. Note that, aside from the limits, our double integral must look as follows:

```
> Doubleint(f(x, y), y, x);
```

$$\int\int \left(2 x^2 y + 5 x - 7 y^3\right) \mathrm{d}y \, \mathrm{d}x \qquad \textbf{(2.1.1.2)}$$

We now take an arbitrary *x* value as indicated by the blue line in the picture below.



On that blue line the variable *y* varies between the y-coordinates of part 2 and part 3 of the boundary.

```
> ymin:=solve(bnd2, y);
  ymax:=solve(bnd3, y);
```

$$ymin := 0$$

$$ymax := -\frac{1}{3} x + 2 \qquad \textbf{(2.1.1.3)}$$

Observe also that within the region of integration the variable *x* ranges between zero and six.

```
> xmin:=0;
  xmax:=6;
```

$$xmin := 0$$

$$xmax := 6 \qquad \textbf{(2.1.1.4)}$$

This results in the double integral:

```
> e1:=Doubleint(f(x, y), y=ymin..ymax, x=xmin..xmax);
```

$$e1 := \int_0^6 \int_0^{-\frac{1}{3}x+2} \left(2x^2 y + 5x - 7y^3\right) \mathrm{d}y\, \mathrm{d}x \quad (2.1.1.5)$$

We first evaluate the inner integral, that is the integral over y.

```
> int_y:=Int(f(x, y), y=ymin..ymax);
  int_y:=value(int_y);
```

$$int\_y := \int_0^{-\frac{1}{3}x+2} \left(2x^2 y + 5x - 7y^3\right) \mathrm{d}y$$

$$int\_y := x^2 \left(-\frac{1}{3}x+2\right)^2 + 5x\left(-\frac{1}{3}x+2\right) - \frac{7}{4}\left(-\frac{1}{3}x+2\right)^4 \quad (2.1.1.6)$$

This result is now integrated over x:

```
> int_x:=Int(int_y, x=xmin..xmax);
```

$$int\_x := \int_0^6 \left(x^2 \left(-\frac{1}{3}x+2\right)^2 + 5x\left(-\frac{1}{3}x+2\right) - \frac{7}{4}\left(-\frac{1}{3}x+2\right)^4\right) \mathrm{d}x \quad (2.1.1.7)$$

```
> double_integral:=value(int_x);
```

$$double\_integral := \frac{276}{5} \quad (2.1.1.8)$$

To obtain this result directly, simply apply the **value** command to the double integral we labeled **e1**.

```
> e1;
```

$$\int_0^6 \int_0^{-\frac{1}{3}x+2} \left(2x^2 y + 5x - 7y^3\right) \mathrm{d}y\, \mathrm{d}x \quad (2.1.1.9)$$

```
> value(e1);
```

$$\frac{276}{5} \quad (2.1.1.10)$$

```
>
```